

F4 Federkonstante statisch

Alle Längen in mm, alle Massen in g

```
In[1]:= uSpiegelscala[x_] := 0.2 + 5 * 10^-4 * x;
```

Anleitung Einführungspraktikum (2007) Seite 11

Lineare Regression

Einführung (2007): Seite 42 Gln. 49, 50 + Seite 42 Gln unter Gln 51

Nollau: Seiten 129...151

Fahrmeir Seiten 104...120

Schenk: Seiten 9 und 17...18

```
In[2]:= linRegScript[data_] := Module[{x, y, a, b, sa, sb, n, xi, xxi, yi, xyi, D, sy},
  x = Transpose[data][[1]];
  y = Transpose[data][[2]];
  n = Length[data];
  xi = Sum[x[[i]], {i, n}];
  xxi = Sum[x[[i]]^2, {i, n}];
  yi = Sum[y[[i]], {i, n}];
  xyi = Sum[x[[i]] * y[[i]], {i, n}];
  D = n * xxi - xi^2;
  a = (n * xyi - xi * yi) / D;
  b = (xxi * yi - xi * xyi) / D;
  sy = Sqrt[Sum[(y[[i]] - a * x[[i]] - b)^2, {i, n}] / (n - 2)];
  sa = sy * Sqrt[n / D];
  sb = sy * Sqrt[xxi / D];
  {a, sa, b, sb}];
```

gewichtete lineare Regression

Einführung (2007) Seite 41 Gln. 43 + Seite 42 Gln. 45

```
In[3]:= WeightedlinRegScript[data_, s_] :=
  Module[{x, y, a, b, sa, sb, n, ni, xi, xxi, yi, xyi, D},
  x = Transpose[data][[1]];
  y = Transpose[data][[2]];
  n = Length[data];
  ni = Sum[1 / s[[i]]^2, {i, n}];
  xi = Sum[x[[i]] / s[[i]]^2, {i, n}];
  xxi = Sum[x[[i]]^2 / s[[i]]^2, {i, n}];
  yi = Sum[y[[i]] / s[[i]]^2, {i, n}];
  xyi = Sum[x[[i]] * y[[i]] / s[[i]]^2, {i, n}];
  D = ni * xxi - xi^2;
  a = (ni * xyi - xi * yi) / D;
  b = (xxi * yi - xi * xyi) / D;
  sa = Sqrt[ni / D];
  sb = Sqrt[xxi / D];
  {a, sa, b, sb}];
```

Nollau : Seiten 152 ... 156

Fahrmeir Seiten 178 ... 182

```
In[4]:= WeightedLinRegNollau[data_, s_] :=
Module[{x, y, a, b, sa, sb, n, ni, xi, xxi, yi, xyi, D, sy},
  x = Transpose[data][[1]];
  y = Transpose[data][[2]];
  n = Length[data];
  ni = Sum[1 / s[[i]]^2, {i, n}];
  xi = Sum[x[[i]] / s[[i]]^2, {i, n}];
  xxi = Sum[x[[i]]^2 / s[[i]]^2, {i, n}];
  yi = Sum[y[[i]] / s[[i]]^2, {i, n}];
  xyi = Sum[x[[i]] * y[[i]] / s[[i]]^2, {i, n}];
  D = ni * xxi - xi^2;
  a = (ni * xyi - xi * yi) / D;
  b = (xxi * yi - xi * xyi) / D;
  sy = Sqrt[Sum[(y[[i]] - a * x[[i]] - b) / s[[i]]^2, {i, n}] / (n - 2)];
  sa = sy * Sqrt[ni / D];
  sb = sy * Sqrt[xxi / D];
  {a, sa, b, sb}];
```

Mittelwert, Standardabweichung der verwendeten insgesamt 32 Massestücke

```
In[5]:= massePlatz1 = {50.169, 50.954, 50.322, 50.719, 50.207, 50.034, 49.996, 50.612};
massePlatz2 = {50.651, 50.598, 49.840, 50.502, 50.333, 50.016, 49.948, 49.806};
massePlatz3 = {50.871, 50.676, 50.251, 49.966, 50.047, 49.970, 49.945, 49.978};
massePlatz4 = {49.922, 49.958, 49.966, 49.562, 49.988, 50.813, 49.968, 50.400};
masse = Join[massePlatz1, massePlatz2, massePlatz3, massePlatz4];
{Mean[masse], StandardDeviation[masse]}
```

```
Out[10]= {50.2184, 0.35794}
```

Messdatenreihe vom 7.1.2015, Meßplatz 2

Anleitung Einführungspraktikum (2007) Seite 11

Die Feder wird schrittweise bis 400 g mit Massestücken von 50 g belastet und mit Hilfe einer Spiegelskala an der Marke M die Auslenkung x gemessen (erste Messreihe). Nach Erreichen der Höchstbelastung von 400 g ist dann die Feder wieder schrittweise zu entlasten (zweite Messreihe).

```
In[11]:= messdaten = {{186.0, 175.5, 165.0, 155.0, 145.5, 135.5, 126.0, 116.0},
  {115.5, 125.0, 136.0, 146.0, 156.0, 165.5, 176.0, 186.0}};
```

```
In[12]:= n1 = Length[messdaten[[1]]];
n2 = Length[messdaten[[2]]];
xlist = Join[Table[i, {i, n1}], Table[i, {i, n2, 1, -1}]];
ylist = Flatten[messdaten];
fitlist = Transpose[{xlist, ylist}]
```

```
Out[16]= {{1, 186.}, {2, 175.5}, {3, 165.}, {4, 155.}, {5, 145.5},
  {6, 135.5}, {7, 126.}, {8, 116.}, {8, 115.5}, {7, 125.},
  {6, 136.}, {5, 146.}, {4, 156.}, {3, 165.5}, {2, 176.}, {1, 186.}}
```

Abschätzung des y-Fehlers aus dem systematischen Restfehler

```
In[17]:= usysResty = uSpiegelscala[ylist]
```

```
Out[17]= {0.293, 0.28775, 0.2825, 0.2775, 0.27275, 0.26775, 0.263,
  0.258, 0.25775, 0.2625, 0.268, 0.273, 0.278, 0.28275, 0.288, 0.293}
```

Lineare Regression

```
In[18]:= fitresult = LinearModelFit[fitlist, x, x];
Print[fitresult["ParameterTable"]];
```

	Estimate	Standard Error	t-Statistic	P-Value
1	195.723	0.225292	868.753	1.58279×10^{-34}
x	-10.0149	0.0446145	-224.476	2.67212×10^{-26}

```
In[20]:= {a, ua, b, ub} = linRegScript[fitlist];
Print["linRegScript Gln.49 50 und 51a : y = a * x + b "];
Print[TableForm[{b, ub}, {a, ua}],
      TableHeadings -> {"b", "a"}, {"Estimate", "StandardError"}]];
linRegScript Gln.49 50 und 51a : y = a * x + b
```

	Estimate	StandardError
b	195.723	0.225292
a	-10.0149	0.0446145

gewichtete lineare Regression (mit systematischem Restfehler)

```
In[23]:= weigthedfitresult = LinearModelFit[fitlist, x, x, Weights -> 1 / (usysResty) ^ 2];
Print[weigthedfitresult["ParameterTable"]];
```

	Estimate	Standard Error	t-Statistic	P-Value
1	195.707	0.233646	837.62	2.6382×10^{-34}
x	-10.0114	0.044768	-223.628	2.81746×10^{-26}

```
In[25]:= {aS, uaS, bS, ubS} = WeightedlinRegScript[fitlist, usysResty];
Print["WeightedlinRegScript Gln.43 ... 45 : y = a * x + b "];
Print[TableForm[{bS, ubS}, {aS, uaS}],
      TableHeadings -> {"b", "a"}, {"Estimate", "StandardError"}]];
WeightedlinRegScript Gln.43 ... 45 : y = a * x + b
```

	Estimate	StandardError
b	195.707	0.156635
a	-10.0114	0.0300122

```
In[28]:= {aN, uaN, bN, ubN} = WeightedlinRegNollau[fitlist, usysResty];
Print["WeightedlinRegNollau pp. 152 - 155 : y = a * x + b "];
Print[TableForm[{bN, ubN}, {aN, uaN}],
      TableHeadings -> {"b", "a"}, {"Estimate", "StandardError"}]];
WeightedlinRegNollau pp. 152 - 155 : y = a * x + b
```

	Estimate	StandardError
b	195.707	0.233646
a	-10.0114	0.044768

gewichtete lineare Regression (mit systematischem Restfehler) VarianceEstimatorFunction-> (1&)

```
In[31]:= weigthedfitresult = LinearModelFit[fitlist, x, x,
      Weights -> 1 / (usysResty) ^ 2, VarianceEstimatorFunction -> (1 &)];
Print[weigthedfitresult["ParameterTable"]];
```

	Estimate	Standard Error	t-Statistic	P-Value
1	195.707	0.156635	1249.44	9.77167×10^{-37}
x	-10.0114	0.0300122	-333.577	1.04454×10^{-28}